# $Id: PKGBUILD 83120 2013-01-27 15:51:39Z pierre $
# Maintainer: Kevin Piche <kevin@archlinux.org>
# Contributor: K. Piche <kpiche@rogers.com>

pkgname=avfs
pkgver=1.0.1
pkgrel=1
pkgdesc="A virtual filesystem that allows browsing of compressed files"
arch=('i686' 'x86_64')
license=('GPL' 'LGPL')
url="http://avf.sourceforge.net/"
depends=('fuse' 'xz' 'perl')
options=('!emptydirs' '!libtool')
source=(http://downloads.sourceforge.net/sourceforge/avf/${pkgname}-${pkgver}.tar.bz2{,.asc})
md5sums=('f0f254d092e54da14bbf3ea22b76da99'
         'd71607de21819f3d67246473f0bc21f3')

build() {
  cd "${srcdir}/${pkgname}-${pkgver}"
  ./configure --prefix=/usr \
              --enable-library \
              --enable-fuse \
              --disable-static
  make
}

package() {
  cd "${srcdir}/${pkgname}-${pkgver}"

  make DESTDIR="${pkgdir}" install
}
